
Temat i cel pracy

W dzisiejszych czasach każdy użytkownik komputera przyzwyczajony jest do kolorowego pulpitu, wysokich rozdzielczości. Dobry program może się nie sprzedawać dobrze, jeśli nie jest przyozdobiony. Użycie grafiki komputerowej znacznie ułatwia pracę z komputerem, użytkownikowi nie sprawia żadnej trudności wskazanie myszą obiektu i kliknięcie go. Dzięki temu, nie każdy człowiek obsługujący komputer musi znać wszystkie polecenia DOSa. 

Celem pracy było wykonanie programu prezentującego efekty graficzne w Turbo Pascalu. Zaplanowałem zbudować go z kilku części głównych takich jak: szybka animacja 2D, możliwości wyświetlania trybów VESA (wysoka rozdzielczość), prosta prezentacja grafiki trójwymiarowej, kilka efektów tworzonych programowo (ładna grafika bez plików zewnętrznych). 

Wielu osobom, które miały do czynienia z systemem DOS, kojarzy się on z trybem tekstowym a jeśli występuje w nim grafika to już najczęściej w słabych rozdzielczościach, z małą głębią koloru. Chciałem żeby użytkownik poznał prawdziwe możliwości programu uruchamianego wyłącznie pod dosem. Korzystając z samych standardów postanowiłem napisać program wykorzystujący wszystkie graficzne możliwości systemu takiego jak Windows.

Turbo Pascal jest dosyć starym narzędziem programistycznym i uważanym przez wielu za przestarzałe. Umożliwia jednak zbudowanie bardzo wielu ciekawych efektów. Turbo Pascal posiada wbudowany kompilator asemblera, co daje prawie nieograniczone możliwości tworzenia programu. Asembler ma również wiele wad, szczególnie jeśli chodzi o złożoność kodu przy użyciu najbardziej nisko poziomowych rozkazów. W rezultacie programowanie z niskiego poziomu bardziej zaawansowanych efektów jest utrudnione i zniechęcające. Potrzeba wielu godzin aby go napisać.

W swojej pracy starałem się uzyskać jak najlepsze efekty jeśli chodzi o wydajność programu, dzięki czemu przy wyłączonym oczekiwaniu na powrót pionowy, program potrafi rysować nawet blisko 500 pełnoekranowych animacji na sekundę. 
Obsługa i Działanie programu

O programie

Program został napisany w kompilatorze firmy Borland: Turbo Pascal 7.0. W kodzie źródłowym znajdują się także wstawki asemblera, które znacznie przyśpieszają niektóre operacje. Program korzysta z pamięci konwencjonalnej czyli maksymalnie z 640KB, które udostępnia DOS. 

Po zakończeniu prac wszystkie kody źródłowe, z których składały się moduły zajmowały 292 KB i miały łącznie około 13200 linii. Program po skompilowaniu zajmuje 74,5KB.

Wymagania systemowe i sprzętowe
Program poprawnie działa w systemach operacyjnych: DOS, Windows95, Windows98, Windows ME, Windows XP. Niestety w Windows 2000 występują błędy. Wymaga zainstalowanego sterownika myszy. W DOS-ie można zainstalować sterownik Imouse.com dodany do katalogu DemoTP.

Aby praca działała poprawnie należy uruchomić ją w systemie, który ma dostępne minimum 300KB wolnej pamięci konwencjonalnej. Program wymaga procesora 32 bitowego, zaczynając od 386DX lub 486DX. Powinien posiadać kartę graficzną o minimalnej pamięci 1MB (tylko do menu Obrazy, reszta wymaga 128KB).

Obsługa


Program składa się z kilku części. Po uruchomieniu sprawdza, czy nie napotkano żadnego błędu, takiego jak brak pliku, brak myszy, ilość wolnej pamięci. Jeśli wystąpiły błędy, zadaje pytanie użytkownikowi czy ma kontynuować program. Następnie wyświetla menu z wyborem języka. Obsługuje wersję angielską i polską. Po wciśnięciu odpowiedniego przycisku obraz ściemnia się, program przechodzi do części głównej, systemowej. Tutaj wczytuje kolejną porcję danych. Teraz potrzebuje obrazków ikon, tła bitmapowego i języka. Tworzy także w pamięci obiekty, które później staną się przyciskami, oknami i podobnymi elementami.

Cały interfejs użytkownika zbudowany jest podobnie do systemu Windows. Napisałem bibliotekę, która zachowuje się podobnie jak system a nazywa się Doors. Dzięki temu nie sprawia żadnych problemów tworzenie większej ilości obiektów, bo nie trzeba każdego oprogramować z osobna. 

Na pulpicie umieszczone są ikony. Każda z nich ma swoje zadanie. Służą do wyświetlania efektów graficznych, wyświetlania obrazków w wysokiej rozdzielczości, wyświetlania grafiki trójwymiarowej, informacji o autorze, opisu do programu,  oraz pokazywania szybkiej animacji (ustawionej domyślnie na 140 ruchomych obiektów);


Dodatkowe funkcje dostępne są na pasku START, wystarczy go kliknąć, a wyświetli się dodatkowe menu. Opcja Programy jest wyłącznie demonstracją modułu obiektowego Doors i po kliknięciu wyświetla kolejny panel z wyborem funkcji. Pomoc, wyświetla plik tekstowy Help.txt, który znajduję się w katalogu DemoTp\. Od Autora jest tym samym co ikona Autor. W Ustawieniach można zmienić niektóre parametry programu, takie jak oczekiwanie na powrót pionowy czy automatyczne obracanie figur w przestrzeni. Menu Uruchom, wyświetla okno, które pozwala na uruchomienie jakiegoś programu. W ostatnim elemencie paska można wyłączyć system, uruchomić ponownie komputer (niestety Windows nie dopuszcza do tego, jednak wszystko działa z poziomu DOS).
Treści programowe
Turbo Pascal


Mikroprocesor wykonując program interpretuje rozkazy zapisane w tzw. kodzie maszynowym, będącym ciągiem bitów zero-jedynkowych. Ze względu na małą czytelność tak zapisanego programu stworzono szereg języków i systemów programowania.
Język definiuje reguły (zasady) zgodnie z którymi powinien być zapisany program, aby możliwe było jego wykonanie, system programowania natomiast służy do tłumaczenia programu na kod maszynowy.
Języki programowania ze względu na sposób złożoności dzieli się na: 

· niskiego (asembler), 

· wysokiego (Turbo Pascal, C/C++, Java, Fortran, Visual Basic, Algol) 

poziomu. W języku niskiego poziomu większość instrukcji odpowiada pojedynczym rozkazom procesora, dlatego też zapis programu w tym języku jest procesem bardzo pracochłonnym. Języki wysokiego poziomu używają sformułowań stosowanych w życiu codziennym, zaś ich instrukcje zastępują kilka czy nawet kilkanaście instrukcji języka niskiego poziomu. 

Asembler – programowanie


Asembler to język niskiego poziomu, za pomocą którego można wydawać bezpośrednie polecenia do procesora. Dzięki niemu można operować na rejestrach i jak najrzadziej odwoływać się do zmiennych. Takie operowanie kilkukrotnie przyśpiesza wszelkie działania. Asembler nie generuje żadnego zbędnego kodu i tylko od programisty zależy jego jakość i wydajność.


Rejestry są to komórki pamięci wewnątrz procesora, które służą do wykonywania określonych operacji. Każda zmienna którą np. dodajemy przekazywana jest najpierw do rejestru, później wykonuje się na niej działania (dodawanie), a następnie znowu trafia do zmiennej. Widać więc, że część zadań można wykonać w ogóle bez korzystania ze zmiennych. Każdy komputer udostępnia kilka rejestrów:

	Nazwa rejestru 32 bitowego
	Nazwa rejestru 16 bitowego
	Górny bajt
	Dolny bajt

	EAX
	AX
	Ah
	Al

	EBX
	BX
	Bh
	Bl

	ECX
	CX
	Ch
	Cl

	EDX
	DX
	Dh
	Dl

	Rejestry przechowujące adresy danych
	Segment
	Offset
	Opis

	ES:DI
	ES
	DI
	Rejestr zapisu danych

	DS:SI
	DS
	SI
	Rejestr odczytu danych



Oprócz rejestrów Asembler udostępnia polecenia, które pozwalają uzyskać szybsze obliczenia niż za pomocą standardowego edytora pascala. Daje np. możliwość bezpośredniego kopiowania danych zawartych w rejestrze DS:Si do ES:Di. Pozwala na kopiowanie po jednym bajcie, po dwóch i po czterech. Taka operacja zajmuje kilkakrotnie mniej taktów procesora niż skopiowanie adresu pamięci do rejestru i z rejestru do pamięci, a jest chyba kilkadziesiąt razy szybsza od kopiowania tablicy przez indeks, np.

   FOR i:=1 TO n DO Dest[i]:=Source[i];

Niektóre polecenia asemblera:

	Polecenie
	Przykład użycia
	Opis

	MOV
	Mov  Ax, 5

Mov  Ax, Bx

Mov  Ax, zmienna

Mov  Ax, ES:[Di]


	Polecenie wpisuje do rejestru podaną wartość, rejestr, zmienną, lub adres pamięci.

	SHR
	Shr  Ax, 3

Shr  Ax, Dl
	Przesunięcie bitowe rejestru o wskazaną ilość miejsc w prawo, tzn. jeśli mamy w Ax warość 128 –bitowo10000000 to po SHR Ax, 3 będzie 00100000, czyli 32. Często stosowane jako szybkie dzielenie przez potęgę dwójki (w przykładzie przez 2^3 czyli 8) (128:8=32) Można też do Dl wpisać jakąś wartość i wtedy przesuwać o ilość wskazaną w dl

	SHL
	Shl  Ax, 2

Shl  Ax, Dl
	Użycie podobne do przesunięcia w prawo, ale działanie odwrotne. Można używać jako mnożenia przez potęgę dwójki

	ADD
	Add  Ax, Bx

Add  Ax, 3
	Dodawanie dwóch rejestrów lub wartości do rejestru. Do Ax zostanie dodana wartość druga

	INC
	Inc  Ax
	Inkrementacja –zwiększenie Ax o 1. Dużo szybsze (4x) od ADD Ax, 1

	DEC
	Dec  Ax
	Zmniejszenie Ax o 1

	SUB
	Sub  Ax, Bx

Sub  Ax, 4
	Odejmowanie. Od Ax odejmuje Bx

	AND
	And  Al, 11110000b

And  Ax, Bx
	Nakładanie maski bitowej, inaczej operacja and, bity ustawione jako 1 zostaną, reszta będzie wyzerowana

	NOT
	Not  Ax
	Zmienia bity 1 na 0 a 0 na 1, wartość odwrotna, np. jeśli Al=128, wynikiem Not Al będzie 127, bo

NOT 10000000=01111111, można stosować jako szybkie odejmowanie wartości od np. 255 (255-128=127; 255-1=254)

	OR
	Or  Ax, Bx

Or  Bx, 200
	Wszystkie bity które były ustawione dla Bx lub Ax będą ustawione, np. 11100000 OR 00000111=11100111. Wynik zwraca w pierwszym wskazanym rejestrze w przykładzie w Ax



	STOSSB

STOSSW

STOSSD
	StosSW
	Do obszaru pamięci w ES:[Di], wpisuje wartość Ah, Ax lub Eax, zależnie od użytego rodzaju instrukcji i zwiększa rejestr Di odpowiednio 1 (Dla instrukcji z zakończeniem SB), 2 (SW) lub 4 (SD)

	REP
	Rep StosW
	Powtarza instrukcję użytą za Rep, np. StosSW Cx razy

	LOOP


	Loop @Poczatek
	Odejmuje od Cx 1 I przeskakuje do początku programu jeśli Cx>0

	CMP
	Cmp Ax, Bx

Je @Rowne

Ja @AxMniejsze
	Porównuje wskazane wartości, wyniki zapisuje za pomocą flag:

Ja   pierwsza wartość <druga wartość

Jb   pierwsza wartość >druga wartość

Jna   pierwsza wartość <=druga wartość

Jnb   pierwsza wartość >=druga wartość

Je   pierwsza wartość =druga wartość

Jz   pierwsza wartość =0

	MOVSB

MOVSW

MOVSD
	MovSB

MovSW

MovSD
	Przenosi dane z adresu DS:[Si] do ES:[Di], oraz do Si i Di dodaje odpowiednio 1, 2 lub 4, dzięki czemu można tą instrukcję wykonywać w połączeniu z REP

	IN
	In    Ah, Dx
	Zwraca wartość portu urządzeń nr Dx do Ah

	OUT


	Out  Ah, Dx
	Zapisuje wartość Ah do portu urządzeń nr Dx

	LES

LDS
	Les Di, Dest

Lds Si,  Source
	Zapisuje do rejestrów danych i zapisu adres pamięci podany przez jakąś zmienną typu pointer



	CLD
	Cld
	Ustawia że do rejestrów zapisu i odczytu będzie dodawana wartość Di i Si podczas użycia StosSB lub MovSB itp..



Żeby używać w Pascalu wstawek asemblera, wystarczy dopisać ASM, zakańcza się poleceniem END;. Poniższy program pokazuje w jaki sposób zrealizować dodawanie asemblerem, nie jest on jednak dobrym zastosowaniem w programie, jednak daje przykład wykorzystania komendy ADD:

VAR a, b:Word

BEGIN

  ReadLn(a, b);

  ASM

    Mov Ax, a

    Mov Bx, b

    Add Ax, Bx

    Mov a, Ax

  END;

  WriteLn(a);

END.
Z przykładu warto zauważyć, że operacje procesor może wykonywać wyłącznie na rejestrach, a nie na zmiennych, nie możemy bezpośrednio dodać Add a, b. Kolejny przykład, trochę bardzie zaawansowany pokazuje użycie instrukcji zapisywania do obszaru pamięci –StosSW, będzie to przyśpieszona do 16 bit instrukcja FillChar16:

PROCEDURE FillChar16(VAR Dest; Size, Data:Word);ASSEMBLER;

ASM

  {POBIERA ADRES DANYCH}

  Les  Di, Dest

  {ZAPISUJE ROZMIAR DANYCH, KTÓRY ZMIENIMY}

  Mov  Cx, Size

  {ZAPISUJE WARTOSC JAKA BEDZIE WYCZYSZCONY BUFOR DANYCh}

  Mov  Ax, Data

  {KOPIA ILOSCI DANYCH}

  Mov  Bx, Cx

  {JESLI DANYCH MAMY NIEPARZYSTA ILOSC, Bx TO ZAPAMIETA}

  And  Bx, 1

  {DZIELENIE DANYCH NA 2, BO BEDZIEMY KOPIOWALI PO 2 BAJTY}

  Shr   Cx, 1

  {PRZY KAZDYM StosSW Di BĘDZIE WIEKSZE O 2}

  Cld

  {POWTARZA Cx RAZY ZAPISYWANIE DO ES:[Di] WARTOSCI Ax}

  Rep  StosSW

  {JESLI BYLA NIEPARZYSTA ILOSC DANYCH PRZYPOMINA JA SOBIE I USTAWIA LICZNIK NA 1}

  Mov  Cx, Bx

  {JESLI LICZNIK=1 KOPIJE 1 BAJT}

  Rep  StosSB 

END;

Dzięki asemblerowi można programować znacznie szybciej niż za pomocą kompilatora pascala. Daje wiele możliwości, jednak pisanie w nim programów jest męczące. Warto pamiętać, żeby używać asemblera wszędzie tam, gdzie jest potrzebna duża prędkość.

Tworzenie grafiki trójwymiarowej

Ogólnie o 3d 
Cała grafika trójwymiarowa jest tworzona poprzez wiele trójkątów, punktów oraz lini. Aby powstał dowolny obiekt trzeba go skonstruować z właśnie takich figur. Każdy wierzchołek obiektu trzeba zapamiętać, oraz zrzutować na płaszczyznę. Np. jeśli chcesz zaprogramować rozgwieżdżone niebo musisz symulować tysiące punktowych gwiazd, tworząc sześcian, potrzebujesz dwunastu trójkątów, z których każde dwa będą tworzyły jedną ścianę. Prawdopodobnie zastanawiasz się, dlaczego nie można od razu narysować sześciu płaszczyzn, być może denerwują Cię też kwadratowe opony w grach samochodowych. Sprawa jest troszkę skomplikowana. Jeśli chodzi o rysowanie płaszczyzn, OpenGL umożliwia takie operacje, jednak programując grafikę z niskiego poziomu np. Pascala, o wiele prościej będzie narysować 2 trójkąty, reszty dowiesz się przy obcinaniu trójkątów oraz sposobie ich wypełniania. Rysowanie obiektów takich jak kule, koła i im podobne jest bardzo czasochłonne. Aby narysować idealną kulę, lub koło musielibyśmy obliczać współrzędne dwuwymiarowe dla każdego z możliwych punktów. Obliczenie współrzędnej każdego punktu wymaga około 100 lub więcej taktów procesora w zależności od rodzaju liczb na których operujemy. Powiększając kulę, zwiększa się ilość potrzebnych wierzchołków więc programiści wolą stosować kwadratowe kształty. Zmniejsza to wymagania sprzętowe.
Cały problem w grafice trójwymiarowej polega na napisaniu odpowiedniego programu, który w przyszłości będzie za nas wykonywał potrzebne prace. Po napisaniu programu z łatwością będziesz mógł tworzyć dowolnie skomplikowane obiekty, tworzyć gry trójwymiarowe za pomocą jednej instrukcji będziesz potrafił zmieniać pozycję kamery.

Jeśli nie lubisz programować tak skomplikowanych kodów spróbuj OpenGL, tam wszystko jest już napisane. Jedyny problem pojawia się z ułożeniem danych w pamięci, jednak po stworzeniu odpowiedniego edytora sprawa zamyka się na narysowaniu i dopasowaniu tekstur.
Jak tworzyć?
Grafika 3d jest po prostu interpretacją, tego co widzimy, wszystko trzeba wyświetlić na monitorze dwuwymiarowym i sprawić efekt przestrzeni. Żeby lepiej zrozumieć w jaki sposób będziemy przeliczali współrzędne rzeczywiste (x, y, z) na dwuwymiarowe (x, y), weź kartkę papieru, narysuj na środku pionową linię po lewej kamerę, a po prawej stronie jakiś prosty obiekt, np.punkt. Połącz go ze środkiem obiektywa. Teraz przeprowadź prostą prosopadlą do ekranu monitora, przecinającą środek obiektywa. Narysuj ostatnią linię, równoległą do monitora, przecinającą punkt. Powinieneś uzyskać trójkąt. Prosto zauważyć, proporcję.

Oznaczmy przez Sy -Środek monitora y' -punkt przecięcia lini łączącej obiekt ze środkiem obiektywa, Py -Punkt pozycja pionowa, Pz - Zagłębienie punktu, O -Odległość kamery od monitora. Powstaje proporcja:

y'/(Py-Sy)=O/Pz

Aby obliczyć współrzędną Y nowo zrzutowango punktu wystarczy przekształcić równanie, co daje:
y'=(Py-Sy)*(O/Pz)

Obracanie bryły polega na przekształceniu każdego punktu w przestrzeni za pomocą odpowiednich wzorów. W Pascalu może to wyglądać tak:

PROCEDURE G3D_RotateVertex(Source:PVertex; KatX, KatY, KatZ:Byte; Count:Word);

VAR I     : Word;

    Temp  : Single;

BEGIN

  IF NOT Assigned(Source) THEN Exit;

  FOR I:=1 TO Count DO

  BEGIN

    {OBROT WOKOL OSI Y :}

    IF KatY>0 THEN

    BEGIN

      Temp:=Source^.Z*G3D_Sin[KatY]+Source^.X*G3D_Cos[KatY];

      Source^.Z:=Source^.Z*G3D_Cos[KatY]-Source^.X*G3D_Sin[KatY];;

      Source^.X:=Temp;

    END;

    {OBROT WOKOL OSI X :}

    IF KatX>0 THEN

    BEGIN

      Temp:=Source^.Y*G3D_Sin[KatX]+Source^.Z*G3D_Cos[KatX];

      Source^.Y:=(Source^.Y*G3D_Cos[KatX]-Source^.Z*G3D_Sin[KatX]);

      Source^.Z:=Temp;

    END;

    {OBROT WOKOL OSI Z :}

    IF KatZ>0 THEN

    BEGIN

      Temp:=Source^.Y*G3D_Sin[KatZ]+Source^.X*G3D_Cos[KatZ];

      Source^.Y:=(Source^.Y*G3D_Cos[KatZ]-Source^.X*G3D_Sin[KatZ]);

      Source^.X:=Temp;

    END;

    {PRZEJSCIE DO NASTEPNEGO WIERZCHOLKA}

    Inc(Source);

  END;

END;

{*************************************************************************}

PROCEDURE G3D_Rotate(Source:PTriangle; KatX, KatY, KatZ:Byte; Count:Word);

VAR I     : Word;

BEGIN

  IF NOT Assigned(Source) THEN Exit;

  {PETLA DLA Count TROJKATOW}

  FOR I:=1 TO Count DO

  BEGIN

    {PRZEKSZTALCA O KAT 3 WIERZCHOLKI}

    G3D_RotateVertex(@Source^.Vertex[0], KatX, KatY, KatZ, 3);

    {PRZECHODZI DO NASTEPNEGO TROJKATA}

    Inc(Source);

  END;

END;
Po odpowiednim przekształceniu brył, należy je wyświetlić. Jest na to kilka sposobów, można na trójkąty nałożyć tekstury, można traktować je jako zbiór 3 punktów, wypełniony jednolitym kolorem trójkąt lub siatka z linii. Jeśli chcemy, np. wyświetlić jednolite trójkąty najlepiej je wyświetlić, z poziomych linii. Metoda ta jest najszybsza.

PROCEDURE S_Move32(Src, Dest:Pointer ;Count : Word); assembler;

ASM

  Mov  Cx, Count

  Mov  Dx, Cx

  And  Dx, 3

  Shr  Cx, 2

  Push Ds

  Lds  Si, Src

  Les  Di, Dest

  Cld

  Db   $F3,$66,$A5

  Mov  Cx, Dx

  Rep  MovSb

  Pop  Ds

END;

PROCEDURE Zamien(S1,S2:PVertex);

VAR P:TVertex;

BEGIN

  S_Move32(S1, @P, SizeOf(TVertex));

  S_Move32(S2, S1, SizeOf(TVertex));

  S_Move32(@P, S2, SizeOf(TVertex));

END;

PROCEDURE G3D_RysujTrojkat(Tr:PTriangle;C:Word);

VAR I, J    : Integer;

    Wa, Wb  : Single;

    P1, P2  : TVertex;

    X, Y, W : Integer;

    StartXa,

    StartXb : Single;

BEGIN

  IF Tr^.Vertex[0].Y>Tr^.Vertex[1].Y THEN Zamien(@Tr^.Vertex[0], @Tr^.Vertex[1]);

  IF Tr^.Vertex[1].Y>Tr^.Vertex[2].Y THEN Zamien(@Tr^.Vertex[1], @Tr^.Vertex[2]);

  IF Tr^.Vertex[0].Y>Tr^.Vertex[1].Y THEN Zamien(@Tr^.Vertex[0], @Tr^.Vertex[1]);

  IF Tr^.Vertex[1].Y<>Tr^.Vertex[0].Y THEN

     Wa:=(Tr^.Vertex[1].X-Tr^.Vertex[0].X-1)/ABS(Tr^.Vertex[1].Y-Tr^.Vertex[0].Y) ELSE Wa:=0.1;

  IF Tr^.Vertex[2].Y<>Tr^.Vertex[0].Y THEN

     Wb:=(Tr^.Vertex[2].X-Tr^.Vertex[0].X)/ABS(Tr^.Vertex[2].Y-Tr^.Vertex[0].Y) ELSE Wb:=0.1;

  StartXa:=Tr^.Vertex[0].X;

  StartXb:=Tr^.Vertex[0].X;

  FOR I:=Integer(Trunc(Tr^.Vertex[0].Y)) TO Integer(Trunc(Tr^.Vertex[1].Y))-1 DO

  BEGIN

    P1.X:=StartXa;

    P1.Y:=I;

    P2.X:=StartXb;

    StartXa:=StartXa+Wa;

    StartXb:=StartXb+Wb;

    IF P1.X>P2.X THEN Zamien(@P1, @P2);

    {$R-}

    X:=Trunc(P1.X);

    Y:=I;

    W:=Trunc(P2.X)-Trunc(P1.X)+1;

    {$R+}

    IF W<1 THEN Continue;

    V_HLine(V_Ekran, X, Y, W, C);

  END;

  IF Tr^.Vertex[2].Y<>Tr^.Vertex[1].Y THEN

     Wa:=(Tr^.Vertex[2].X-Tr^.Vertex[1].X)/ABS(Tr^.Vertex[2].Y-Tr^.Vertex[1].Y) ELSE Wa:=0.01;

  StartXa:=Tr^.Vertex[1].X;

  FOR I:=Integer(Trunc(Tr^.Vertex[1].Y)) TO Integer(Trunc(Tr^.Vertex[2].Y)-1) DO

  BEGIN

    P1.X:=StartXa;

    P2.X:=StartXb;

    StartXa:=StartXa+Wa;

    StartXb:=StartXb+Wb;

    IF P1.X>P2.X THEN Zamien(@P1, @P2);

    {$R-}

    X:=Trunc(P1.X);

    Y:=I;

    W:=Trunc(P2.X)-Trunc(P1.X)+1;

    {$R+}

    IF W<1 THEN Continue;

    V_HLine(V_Ekran, X, Y, W, C);

  END;

END;

Podany przykład korzysta z biblioteki Graph10e, obsługującego m. in. linie poziome.

VESA – Tworzenie szybkiej grafiki


Vesa jest standardem graficznym, który kilka lat temu był bardzo popularny. Nawet dzisiejsze karty graficzne działają zgodnie z nim. Pozwala na tworzenie prostych, niezbyt szybkich animacji w porównaniu z dzisiejszymi kartami graficznymi. Jest kilkanaście razy wolniejszy od zwykłej pamięci, dlatego podczas tworzenia efektów trzeba jak najrzadziej odwoływać się do Biosu karty. Problem można rozwiązać za pomocą dodatkowego bufora pamięci –pobranego z pierwszego megabajta dostępnej pamięci konwencjonalnej. Ma to dosyć dużą wadę, zużywa potrzebne zasoby do zapamiętania obrazków, danych figur i innych. Aby nie zrezygnować z szybkości oraz ze zbyt dużej ilości wolnej pamięci można pójść na kompromis, zmniejszyć rozdzielczość w której będzie pracował program. 

Cala operacja graficzna będzie polegała na wyczyszczeniu bufora pamięci, narysowaniu w nim kolejnej klatki animacji i wyświetleniu wszystkiego na ekranie, najlepiej podczas powrotu pionowego monitora, dzięki czemu zniwelujemy efekt migotania obrazu. Pojawia się kolejny problem. Aby porozumieć się z Biosem mamy do dyspozycji mały wycinek pamięci o rozmiarze 64 KB rozpoczynający się adresem $A000:0. Ponieważ wybrałem tryb 10Eh czyli 32x200x16BPP potrzebujemy przesłać do karty graficznej troszkę więcej niż 64KB, bo 320x200x2 bajtów czyli 128000 bajtów (prawie 2 razy tyle co umożliwia BIOS). Trzeba więc podzielić operację na 2 części. Najpierw przesyłamy 64KB, potem to co pozostało 128000-65536 bajtów. Ponieważ przesyłanie odbywa się bardzo wolno, dobrze napisać przyspieszoną procedurę kopiowania bufora pamięci pod inny adres. W moim module jest to S_Move32 –procedura przyśpieszona 32 bitowymi instrukcjami assemblera.

Rysowanie prostych elementów


Kiedy korzystamy z trybu graficznego VESA nie mamy żadnych dostępnych bibliotek pozwalających na rysowanie ani linii, ani pikseli. Wszystko trzeba zaprogramować od początku. Ma to dużą zaletę. Jeśli wszystko zrobimy dobrze w asemblerze, nasze procedury będą o wiele szybsze niż te, które razem z Pascalem dostarczono w module Graph.


Rysowanie piksela można przyśpieszyć przy wyznaczaniu jego adresu pamięci w buforze. Aby obliczyć który bajt z kolei mamy zaświecić wcale nie musimy używać operacji mnożenia. Kiedy wyznaczamy pozycję w buforze za pomocą dwóch zmiennych x, y zazwyczaj operacja wykonywana jest x+y*szerokość_ekranu. Jeśli znamy rozdzielczość poziomą możemy użyć przesunięcia bitowego (SHL), np. 320=256+64 = Y SHL 8+Y SHL 6=Y*szerokość_ekranu –ma to oczywiście wadę, jeśli zaprogramujemy rysowanie szybkich pikseli, procedura nie będzie standardowa i dopasowana do innych rozdzielczości. Za to będzie kilkukrotnie szybsza od standardowej. Rysowanie linii jest trochę bardziej skomplikowane, wymaga dużej znajomości tematu. Można ją rysować bez operacji dzielenia i mnożenia, w Internecie można znaleźć wiele szybkich do tego celu algorytmów. Kolejną ważną operacją jest rysowanie poziomych linii, ponieważ może to znacznie przyśpieszyć grafikę. Np. zamiast rysować 100 pikseli obok siebie, można narysować linię, co będzie kilkukrotnie szybsze, ponieważ nie wymaga obliczania pozycji każdego piksela w buforze.


Przykładowe procedury rysujące piksel oraz linię poziomą (Wymagają opcji kompilatora {$G+}): 

FUNCTION V_ObetnijHLine(VAR X1, Y1, W:Integer):Boolean;

BEGIN

  V_ObetnijHLine:=TRUE;

  IF (Y1<DefaultRect.Y) OR (Y1>=DefaultRect.Y+DefaultRect.H) OR

     (X1>=DefaultRect.X+DefaultRect.W) OR (W<=0) OR (X1+W<=DefaultRect.X) THEN Exit;

  IF X1<DefaultRect.X THEN


  BEGIN

      IF X1+W>DefaultRect.X+DefaultRect.W THEN W:=DefaultRect.W

      ELSE



    W:=W+(X1-DefaultRect.X);

      X1:=DefaultRect.X;



END ELSE

    IF X1+W>DefaultRect.X+DefaultRect.W THEN




W:=DefaultRect.X+DefaultRect.W-X1;

  V_ObetnijHLine:=FALSE;

END;

{*************************************************************************}

PROCEDURE V_HLine(Buf:TEkran10Eh; X, Y, W:Integer;C:Word);

VAR Ekr1, Ekr2:Pointer;

BEGIN

  IF V_ObetnijHLine(X, Y, W) THEN Exit;

  Ekr1:=Buf.Ekr1;

  Ekr2:=Buf.Ekr2;

  ASM

    Mov  Ax, Y

    Cmp  Ax, 100

    Ja   @Drugi

    Les  Di, Ekr1

    Jmp  @Dalej

    @Drugi:

      Les  Di, Ekr2

      Sub  Ax, 100

    @Dalej:

      Shl  Ax, 7

      Add  Di, Ax

      Shl  Ax, 2

      Add  Di, Ax

      Add  Di, X

      Add  Di, X

      Mov  Cx, W

      Mov  Bx, Cx

      And  Bx, 1

      Shr  Cx, 1

      Mov  Ax, C

      Db   $66

      Shl  Ax, 16

      Mov  Ax, C

      Cld

      Db   $F3, $66, $AB

      Mov  Cx, Bx

      Rep  StosW

   END;

END;

{*************************************************************************}

PROCEDURE V_Pix(Buf:TEkran10Eh; X, Y:Integer;C:Word);

VAR Ekr1, Ekr2:Pointer;

BEGIN

  IF (X<DefaultRect.X) OR (X>=DefaultRect.X+DefaultRect.W) OR

     (Y<DefaultRect.Y) OR (Y>=DefaultRect.Y+DefaultRect.H) THEN Exit;

  Ekr1:=Buf.Ekr1;

  Ekr2:=Buf.Ekr2;

  ASM

    Mov  Ax, Y

    Cmp  Ax, 100

    Ja   @Drugi

    Les  Bx, Ekr1

    Jmp  @Dalej

    @Drugi:

      Les  Bx, Ekr2

      Sub  Ax, 100

    @Dalej:

      Shl  Ax, 7

      Add  Bx, Ax

      Shl  Ax, 2

      Add  Bx, Ax

      Add  Bx, X

      Add  Bx, X

      Mov  Ax, C

      Mov  ES:[Bx], Ax

   END;

END;

{*************************************************************************}
Procedury działają dla zdefiniowanego bufora ekranowego jako:

TYPE

  TEkran10Eh=RECORD

    Ekr1 : Pointer;

    Ekr2 : Pointer;

  END;

Gdzie Ekr1 jest wskaźnikiem na pierwszą część ekranu a Ekr2 na drugą. Deklaracja pamięci odbywa się za pomocą instrukcji:

FUNCTION V_GetBuf(VAR Buf:TEkran10Eh):Byte;

BEGIN

  Buf.EKR1:=NIL;

  Buf.EKR2:=NIL;

  V_GetBuf:=E_OK;

  IF MemAvail<128000 THEN

  BEGIN

    V_GetBuf:=E_PAMIEC;

    Exit;

  END;

  GetMem(Buf.EKR1, 64000);

  GetMem(Buf.EKR2, 64000);

  V_Ekran.EKR1:=Buf.EKR1;

  V_Ekran.EKR2:=Buf.EKR2;

END;

Moduł Graph10E, który zajmuje się obsługą trybu VESA zawiera jeszcze inne procedury, które umożliwiają wczytywanie obrazków, wyświetlanie przezroczystych linii, wyświetlaniem tekstu, oczekiwaniem na powrót pionowy i wiele innych.
Sposób realizacji

Ogólnie

W pracy wykorzystałem wiele różnych struktur danych, które pomogły np. w tworzeniu nieograniczonej liczby obiektów. Oprócz tego próbowałem zaimplementować algorytmy naśladujące proste efekty. Rozdział ten omawia niektóre z rozwiązań, zastosowanych w programie.
Efekt ognia


Jest to prosty efekt pracujący w trybie 320x200x256 kolorów. Na początku ustawia się paletę kolorów, tak aby największy kolor był najjaśniejszy (najgorętszy). Dalsze kolory przechodzą powoli do czerni.

Komputer losuje dwie dolne linie i przypisuje im wartości z przedziału ok. 155-255. Następnym krokiem jest przeglądanie całego ekranu od lewego górnego piksela aż do ostatniego prawego i przypisanie mu wartości średniej czterech sąsiadujących kolorów. Dzięki temu obraz stanie się bardziej wygładzony oraz powstaną języki ognia. Jeśli wartość ta jest większa niż 0 (od koloru czarnego) wtedy zmniejsza się ją o 1, dzięki czemu uzyskuje się efekt zciemniania.
Gwiazdy


Gwiazdy polegają na przekształceniu trzech współrzędnych ekranowych X, Y, Z na dwuwymiarowe. Współrzędne 3 wymiarowe są przybliżane do obserwatora, gwiazda, która doleci do samego początku trafia na sam koniec. Dzięki temu efekt jest bardziej płynny.

Obrazy


Są to bitmapy 16 lub 24 bitowe o dowolnej rozdzielczości. Procedura wczytuje nagłówek pliku bitmapy, a następnie pobiera kolory wszystkich kolejnych pikseli. Bitmapa jest zapisana od tyłu, tzn. na początku pliku znajduje się lewy dolny piksel, na końcu prawy górny. Poniżej znajduje się typ nagłówka bitmapy, wskazuje jakie informacje można z niej uzyskać.
TBMPCaption    = RECORD

    BM           : Word;

    Size         : Longint;

    rezerw       : Longint;

    obraz_offset : Longint;

    info         : Longint;

    Width        : Longint;

    Height       : Longint;

    LPO          : Word;

    BPP          : Word;

    kompresja    : Longint;

    Size_Obr     : Longint;

    HDPI         : Longint;

    VDPI         : Longint;

    Colors       : Longint;

    UColors      : Longint;

  END;
Efekt Star Wars


Efekt ten wykorzystuje wcześniej opisane gwiazdy. Różnica polega na tym, że wyświetla jeszcze płaszczyznę, na którą nałożona jest tekstura. Płaszczyzna porusza się od obserwatora w stronę głębi kosmosu sprawiając efekt poruszania w przestrzeni.
Figury trójwymiarowe


Są tworzone w oparciu o mój moduł G3d. Pozwala on na wczytanie projektu dowolnie dużej figury z pliku tekstowego, transformację, obrót oraz renderowanie obiektu. Zawiera bloki instrukcji, które pozwalają użytkownikowi jedynie skupić się na konstrukcji świata 3d.
Szybkie animowane obrazki


Działają w oparciu o moduł Graph10E, który pozwala na kopiowanie animacji 32 bitowo. Zapewnia dużą szybkość, w programie zastosowałem procedurę, rysującą 140 różnych zdjęć w różnych fazach animacji.
Zasada działania systemu Doors

Informacje wstępne


System Doors jest zbiorem modułów, które pozwalają na tworzenie bardzo skomplikowanych elementów. Symuluje działanie systemu operacyjnego. Ułatwia tworzenie okien, przycisków, pól tekstowych, etykiet, paneli i wielu innych obiektów. Pozwala na obsługiwanie zdarzeń wywołanych sprzętowo i programowo. Budowanym obiektom jest dynamicznie przydzielana pamięć, dzięki czemu ich ilość jest ograniczona tylko wolną pamięcią konwencjonalną.

Budowa systemu

Ogólnie


Doors oparty jest na trzech głównych modułach oraz kilkunastu wspomagających go modułach standardowych takich jak: obsługa grafiki, myszy, klawiatury, proste operacje. Oprócz tego do modułu Doors można dołączać inne moduły pomocnicze, specyficzne dla każdego obiektu. Moduły pomocnicze to np. Forms, Edits, Panels, Labels, Buttons –które obsługują odpowiednio okna, pola tekstowe, panele, etykiety i przyciski.

Moduł Doors


Moduł ten obsługuje wszystkie operacje związane z tworzeniem obiektów, przydzielaniem im pamięci i dodawaniem ich do listy wszystkich obiektów aktualnie istniejących. Pozwala także na dodawanie zdarzeń, które będą wywoływane np. po najechaniu myszką nad przycisk. Jest modułem podstawowym, jądrem systemu Doors.


Spis funkcji i procedur ogólnych:

{TWORZY OBIEKT GLOWNY, NADRZEDNY}

FUNCTION D_CreateObject(VAR Obj:PObject):Byte;

{TWORZY OBIEKT PODRZEDNY}

FUNCTION D_CreateChild(Parent:PObject;VAR Child:PObject):BYTE;

{USUWA OBIEKT}

FUNCTION D_DeleteObject(VAR Obj:PObject):Byte;

{ZWALNIA CALA PAMIEC ZABRANA PRZEZ SYSTEM}

FUNCTION D_FreeAllMemory:Byte;

{NOWE ZDARZENIE -WYWOLYWANE AUTOMATYCZNIE}

FUNCTION D_NewEvent(EvObj:PEventsObj; Proc:TProcedure;Event:Byte):PEventsProc;

Moduł Doors_EV


Jego zadaniem jest obsługa wszelkich zdarzeń (Events), które zostały wywołane. Odpowiada za pobranie stanu myszki oraz klawiatury, sprawdzenie i przekazanie obiektom informacji czy nad nimi znajduje się kursor myszki, czy został kliknięty obiekt, jaki klawisz na klawiaturze został wciśnięty.


Moduł sprawdza też, który z obiektów jest aktualnie na wierzchu, aktywuje obiekty, które znajdują się pod innymi jeśli został wskazany myszką. Sprawuje funkcję komunikatora z użytkownikiem.


Spis funkcji i procedur:

{POBIERA WSZYSTKIE ZDARZENIA}

FUNCTION DE_GetEvents:Byte;

{SPRAWDZA CZY OBECNIE NA OBIEKT Obj DZIALA JAKIES ZDARZENIE}

FUNCTION DE_ObjectEvent(Obj:PObject; Event:Byte):Boolean;

{WYKONUJE PROCEDURY SKOJARZONE ZE ZDARZENIAMI}

PROCEDURE DE_ExecuteEvents(Event:PEventsObj);

{}

PROCEDURE DE_ExecuteKeyBoardEvents(Event:PEventsObj);

{*************************************************************************}

PROCEDURE DE_SetEvents;

PROCEDURE DE_DeleteEvents;
Modul Doors10E


Modul oparty jest na trybie graficznym zgodnym ze standardem VESA (10Eh). Jest to tryb o rozdzielczości 320x200x16BPP. Obsluguje operacje graficzne potrzebne do wyświetlenia wszystkich obiektów na ekranie. Oparty jest na innym module Graph10E, który zostal napisany specjalnie z myślą o przyśpieszeniu wyświetlania grafiki. Zastowsowalem w nim polecenia 32 bitowe asemblera, dzięki czemu pomimo 16 bitowego kompilatora Pascala można uzyskać efekt kopiowania pamięci dwukrotnie szybsze.

Modul Doors10E zawiera tylko ogólne procedury rysujące, nie znajdują się w nim specjalistyczne procedury do obsługi każdego obiektu, lecz jedynie do przyśpieszonego wyświetlania najczęściej używanych elementów (takich jak prostokąt, prostokąt z przejściem kolorowym, wypukłość, wklęsłość, itp…)


Obiekty rysowane są w pewnym porządku, najpierw obiekty główne (rodzice), później mniej znaczące (dzieci), dzięki czemu na oknie znajdują się przyciski, jednak żaden pod nim. Obiekty są także posortowane wedlug priorytetów, tzn. jeśli ostatnio kliknęliśmy jakiś obiekt, będzie rysowany na końcu, dzięki czemu sprawi wrażenie, że jest nad innymi obiektami –na wierzchu.

Procedury i funkcje modulu Doors10E:

{RYSUJE WESZYSTKIE OBIEKTY}

PROCEDURE D10_DrawAll(Buf:TEkran10Eh);

{RYSUJE LISTE OBIEKTOW WSKAZANA PRZEZ ZMIENNA}

PROCEDURE D10_DrawList(Buf:TEkran10Eh; Lista:PObject; X, Y:Integer);

{RYSUJE POJEDYNCZY OBIEKT}

PROCEDURE D10_DrawObject(Buf:TEkran10Eh; Obj:PObject; AddX, AddY:Integer);

{RYSUJE KURSOR MYSZY}

PROCEDURE D10_DrawCursor(Buf:TEkran10Eh; Picture:TPicture);

{RYSUJE TLO -JEDNOLITE}

PROCEDURE D10_DrawBackground(Buf:TEkran10Eh; Alpha: Byte);

{RYSUJE TLO Z PRZEJSCIEM KOLOROWYM}

PROCEDURE D10_DrawBackground2(Buf:TEkran10Eh; Alpha: Byte;C1, C2:Word);

{RYSUJE OBRAMOWANIE}

PROCEDURE D10_DrawBorder(Buf:TEkran10Eh; Typ:Byte; Alpha:Byte);

{RYSUJE OBIEKT ROZDZIELAJACY – WYPUKLA LINIA}

PROCEDURE D10_DrawSeparator(Buf:TEkran10Eh; Typ:Byte);

Podsumowanie


Podczas tworzenia programu spotkałem się z wieloma problemami, np. jakich struktur danych użyć, jak przyśpieszyć grafikę, w jaki sposób zbudować obiekty, które będą reagowały na zdarzenia, dzięki czemu poznałem nowe rozwiązania. Nauczyłem się programować rekurencyjne procedury, poznałem struktury takie jak lista, drzewo, kolejka, dowiedziałem się o podstawowych zasadach tworzenia grafiki trójwymiarowej oraz zdobyłem większe doświadczenie  w tej dziedzinie informatyki.


Myślę, że zrealizowałem znaczną większość celów które sobie postawiłem. Program który napisałem może być pomocny dla wszystkich osób zaciekawionych tworzeniem grafiki, szczególnie z poziomu DOS. Moduły, potrzebne do pracy można wykorzystać w innych pracach, dzięki czemu powstaną znacznie szybciej i będą bardziej funkcjonalne, a po niewielkich zmianach można np. używać system Doors w DirectX pod Delphi.

